



PRODUCT NAME

TM58XX

TITLE

TM58XX 如何使用 TMRO 中斷

APPLICATION NOTE

- 1.簡介
- 2.DEMO 程式
- 3.應用線路圖

簡介

在當前的許多應用程式中，往往需作定時處理。而用 TMR0 中斷來對定時進行處理，往往會簡化程式設計。針對此目的，我們介紹以下 TM58 系列 MCU（以 TM58P10 為例）如何來使用 TMR0 中斷。在使用 TMR0 中斷時，首先要用到 TM58 系列 IC 的 ADVANCE（高級）模式。因為只有在高級模式才提供中斷功能。在該模式下，TM58P10 的位址 3FEH 為中斷向量地址，TM58P20 的地址 7FEH 為中斷向量地址。與 TMR0 中斷有關的特殊功能暫存器有兩個：中斷遮罩暫存器 IRQM(位址為 21H)和中斷標誌暫存器 IRQF（位址為 22H），其具體定義如下：


IRQM：

位	符號	描述
7	INTM	整體允許位： 1：允許 0：禁止
6~3	----	
2	EXINTM	外部中斷允許位： 1：允許中斷 0：禁止中斷
1	----	
0	TMR0M	TMR0 中斷允許位： 1：允許中斷 0：禁止中斷

IRQF：

位	符號	描述
7~3	----	
2	EXINTF	外部中斷標誌： 1：外部中斷產生的外部中斷請求
1	----	
0	TMR0F	TMR0 中斷標誌： 1：TMR0 計數器溢出產生的中斷請求

工作模式的選擇及看門狗 WDT 的硬體的設定位需在 IC 燒錄時在 CONFIG WORD 中設定，CONFIG WORD 如下圖所示：



NOTE1 :

在程式中應注意進入中斷時變數的保存及出中斷時變數的恢復。

NOTE2 :

當有外部中斷時，將由硬體置位元中斷標誌暫存器的 EXINTF 位元（EXINTF 位元為硬體設置位元，由軟體清零），為避免進入中斷死迴圈，必須在退出中斷副程式時用軟體清除該標誌位元。系統在進入中斷時會自動關中斷，在出中斷時由 RETI 指令自動開中斷。

下面是 TM58P10 使用 TMR0 100uS 1mS 2mS 4mS 1S 中斷時的 DEMO 程式和線路圖：程式的功能見相應程式中的描述。

```

;=====
;模式 (mode):高級 (advanced)
;晶振：4MHZ
;看門狗 (WATCHDOG)：禁止 (DISABLE)
;功能:tmr0 100uS 中斷程式
;系統利用 tmr0 中斷產生一個週期為 200uS 的方波,該方波信號
;通過 portb,0 埠輸出
;
;
;=====
;特殊功能暫存器定義
;=====
indf      equ    00h
tmr0     equ    01h
pc       equ    02h
status   equ    03h
fsr      equ    04h
porta    equ    05h
portb    equ    06h
wakeup   equ    20h
irqm     equ    21h
irqf     equ    22h
;
;
;=====
;目的暫存器控制位元定義
;=====
w        equ    00h
f        equ    01h
;
;
;=====
;狀態暫存器(status)位元定義
;=====
c        equ    00h
dc       equ    01h
z        equ    02h
pd       equ    03h
to       equ    04h
pa0      equ    05h
pa1      equ    06h
pa2      equ    07h
;
;
;=====
;中斷遮罩暫存器(irqm)位定義

```

```

;=====
intm          equ    07h
exintm       equ    02h
tmr0m        equ    00h
;
;
;=====
;中斷請求標誌暫存器 (irqf)位元定義
;=====
tmr0f        equ    00h
exintf       equ    02h
;
;
;=====
;select 暫存器位定義
;=====
ps0          equ    00h
ps1          equ    01h
ps2          equ    02h
psa          equ    03h
rte          equ    04h
rts          equ    05h
;
;
;=====
;WAKE_UP 暫存器位定義
;=====
wdts         equ    07h
wue          equ    06h
eis          equ    05h
puh3         equ    03h
puh2         equ    02h
puh1         equ    01h
puh0         equ    00h
;
;
;=====
;位置定義
;=====
a_temp       equ    0bh
status_temp  equ    0ch
portb_buf    equ    0dh
;
;
;=====
;portb 埠      I/O 定義
;=====
wave         equ    00h
;
;
;=====

```

```

;=====
;=====
;program start here
;=====
;=====
org    3ffh    ;重定向量地址
goto   main
org    3feh    ;中斷副程式入埠位址
goto   time_int
org    000h    ;程式在 ROM 中的起始地址
;
;
;=====
;MAIN PROGRAM
;=====
main::;初始化
    clrm    portb    ;portb 埠輸出"00h"
    clrm    portb_buf    ;portb 埠輸出緩衝器清零
    clra
    iodir    portb    ;portb 埠設為輸出
    movla   b'00001000'
    select
                ;tmr0 採用內部時鐘計數,預除頻器分配給 wdt,
                ;除頻比為 1:1

    movla   b'10000001'
    movam   irqm    ;開放 tmr0 中斷功能
    movla   9ch
    movam   tmr0    ;給 tmr0 預設值"9ch"(156),使 tmr0 計數 100 即溢出
;
;
start::;主迴圈
    nop
    nop
    nop
    nop
    nop
    goto   start
;
;=====
;tmr0 中斷副程式
;=====
time_int:
    movam   a_temp    ;將累加器內容送入 a_temp 中
    swapm   status,w    ;將狀態暫存器 status 高低半位元組內容
                    ;相互交換結果送累加器中
    movam   status_temp    ;最後將結果送入 status_temp 中
    movla   0aah    ;給 tmr0 初始值"0aah"(156+7+7),使
                    ;tmr0 每 100uS(256-156=100)中斷一次,

```

;為補償 lgoto time_int~movam tmr0 6 條
 ;指令(耗時 7 個指令週期)及硬體回應中斷所消耗的
 ;時間(耗時 7 個指令週期),給 tmr0 重裝初值時多加
 ;了 14 個計數值

```

movam tmr0
comm portb_buf,f
movm portb_buf,w
andla 01h
movam portb
end_int:
bcm irqf,tmr0f ;清 tmr0 中斷請求標誌
swapm status_temp,w ;將 status_temp 暫存器高低半位元組內容相互交換
;結果送累加器中
movam status ;累加器內容送狀態暫存器 status,以恢復進中
;斷前的 status 狀態
swapm a_temp,f ;將 a_temp 的高低半位元組內容相互交換,
;結果送入 a_temp 中
swapm a_temp,w ;將 a_temp 的高低半位元組內容相互交換,
;結果送入累加器中,以恢復進中斷前的累加器內容
reti ;中斷返回，執行該指令後，程式將自動開中斷
;
;
;=====
end ;程式結束
;
;
;=====

```

;範例

```

;=====
;系統接電後會在 portb,0 埠輸出一個週期為 200uS 的方波,該
;方波是通過 tmr0 中斷來實現的,tmr0 每 100uS 中斷一次.由於
;中斷回應過程以及在中斷副程式中對累加器,狀態暫存器的保
;護和 tmr0 的重裝初值都需要消耗時間(該部分時間共為 14 個指
;令週期),故副程式中對此部時間作了相應補償,補償方式為:
;給 tmr0 多裝 14 個初值,使其少記 14 個數。

```

```

;=====
;
;
;模式 (mode):高級 (advanced)
;晶振：4MHZ
;看門狗 (WATCHDOG)：禁止 (DISABLE)
;功能:tmr0 1mS 中斷程式
;系統利用 tmr0 中斷產生一個週期為 2mS 的方波,該方波信號
;通過 portb,0 埠輸出

```

```

;
;
;=====
;特殊功能暫存器定義
;=====
indf          equ    00h
tmr0          equ    01h
pc            equ    02h
status        equ    03h
fsr           equ    04h
porta         equ    05h
portb         equ    06h
wakeup        equ    20h
irqm          equ    21h
irqf          equ    22h
;
;
;=====
;目的暫存器控制位元定義
;=====
w             equ    00h
f             equ    01h
;
;
;=====
;狀態暫存器(status)位元定義
;=====
c             equ    00h
dc            equ    01h
z             equ    02h
pd            equ    03h
to            equ    04h
pa0           equ    05h
pa1           equ    06h
pa2           equ    07h
;
;
;=====
; 中斷遮罩暫存器(irqm)位元定義
;=====
intm          equ    07h
exintm        equ    02h
tmr0m         equ    00h
;
;
;=====
;中斷請求標誌暫存器 (irqf)位元定義
;=====
tmr0f         equ    00h
exintf        equ    02h

```

```

;
;
;=====
;select 暫存器位定義
;=====
ps0      equ    00h
ps1      equ    01h
ps2      equ    02h
psa      equ    03h
rte      equ    04h
rts      equ    05h
;
;
;=====
;WAKE_UP 暫存器位定義
;=====
wdts     equ    07h
wue      equ    06h
eis      equ    05h
puh3     equ    03h
puh2     equ    02h
puh1     equ    01h
puh0     equ    00h
;
;
;=====
;位置定義
;=====
a_temp   equ    0bh
status_temp equ 0ch
portb_buf equ 0dh
;
;
;=====
;portb 埠    I/O 定義
;=====
wave     equ    00h
;
;
;=====
;program start here
;=====
org      3ffh      ;重定向量地址
goto    main
org      3feh      ;中斷副程式入埠位址
goto    time_int
org      000h      ;程式在 ROM 中的起始地址
;
;
;=====

```

```

;=====
;MAIN PROGRAM
;=====
main::;初始化
    clrm    portb        ;portb 埠輸出"00h"
    clrm    portb_buf    ;portb 埠輸出緩衝器清除為零
    clra
    iodir   portb        ;portb 埠設為輸出
    movla   b'00000010'
    select
            ;tmr0 採用內部時鐘計數,預除頻器分配給 tmr0,
            ;除頻比為 1:8

    movla   b'10000001'
    movam   irqm         ;開放 tmr0 中斷功能
    movla   83h
    movam   tmr0        ;給 tmr0 預設值"83h"(131),使 tmr0 計數 125 即溢出
;
;
;
start::;主迴圈
    nop
    nop
    nop
    nop
    nop
    goto    start
;
;
;=====
;tmr0 中斷副程式(每 1mS 中斷次)
;=====
    time_int:
    nop            ;時間補償指令
    nop            ;時間補償指令
    movam   a_temp ;將累加器內容送入 a_temp 中
    swapm   status,w ;將狀態暫存器 status 高低半位元組內容
                ;相互交換結果送累加器中

    movam   status_temp ;最後將結果送入 status_temp 中進
    movla   85h        ;給 tmr0 初始值"85h"(131+1+1),使
                ;tmr0 每 1mS((256-131)*8=1000uS)中斷一次,
                ;為補償 lgoto time_int~movam tmr0 9 條
                ;指令(耗時 9 個指令週期)及硬體回應中斷所消耗的
                ;時間(耗時 7 個指令週期),因 tmr0 除頻比為 1:8,給
                ;tmr0 重裝初值時多加了 2(2=16/8)個計數值

    movam   tmr0
    comm    portb_buf,f
    movm    portb_buf,w

```

```

    andla    01h
    movam   portb
end_int:
    bcm     irqf,tmr0f    ;清 tmr0 中斷請求標誌
    swapm   status_temp,w ;將 status_temp 暫存器高低半位元組內容相互交換
                                ;結果送累加器中
    movam   status       ;累加器內容送狀態暫存器 status,以恢復進中
                                ;斷前的 status 狀態
    swapm   a_temp,f     ;將 a_temp 的高低半位元組內容相互交換,
                                ;結果送入 a_temp 中
    swapm   a_temp,w     ;將 a_temp 的高低半位元組內容相互交換,
                                ;結果送入累加器中,以恢復進中斷前的累加器內容
    reti    ;中斷返回，執行該指令後，程式將自動開中斷
;
;
;=====
    end          ;程式結束
;
;
;=====

```

;範例

```

;=====
;系統上電後會在 portb,0 埠輸出一個週期為 2mS 的方波,該
;方波是通過 tmr0 中斷來實現的,tmr0 每 1mS 中斷一次.由於
;中斷相應過程以及在中斷副程式中對累加器,狀態暫存器的保
;護和 tmr0 的重裝初值都需要消耗時間(該部分時間共為 16 個指
;令週期),故副程式中對此部時間作了相應補償,補償方式為:
;因除頻比為 1:8,故給 tmr0 多裝 2 個初值,使其少記 2 個數。
;=====

```

;模式 (mode):高級 (advanced)

;晶振：4MHZ

;看門狗 (WATCHDOG)：禁止 (DISABLE)

;功能:tmr0 2mS 中斷程式

;系統利用 tmr0 中斷產生一個週期為 4mS 的方波,該方波信號

;通過 portb,0 埠輸出

;

;

;

;特殊功能暫存器定義

;

indf equ 00h

tmr0 equ 01h

pc equ 02h

```

status      equ    03h
fsr         equ    04h
porta      equ    05h
portb      equ    06h
wakeup     equ    20h
irqm       equ    21h
irqf       equ    22h
;
;
;
;=====

```

```

;目的暫存器控制位元定義
;=====

```

```

w          equ    00h
f          equ    01h
;
;
;
;=====

```

```

;狀態暫存器(status)位元定義
;=====

```

```

c          equ    00h
dc         equ    01h
z          equ    02h
pd         equ    03h
to         equ    04h
pa0        equ    05h
pa1        equ    06h
pa2        equ    07h
;
;
;
;=====

```

```

; 中斷遮罩暫存器(irqm)位定義
;=====

```

```

intm       equ    07h
exintm     equ    02h
tmr0m      equ    00h
;
;
;
;=====

```

```

;中斷請求標誌暫存器 (irqf)位元定義
;=====

```

```

tmr0f      equ    00h
exintf     equ    02h
;
;
;
;=====

```

```

;select 暫存器位定義
;=====

```

```

ps0        equ    00h
ps1        equ    01h

```

```

ps2      equ    02h
psa      equ    03h
rte      equ    04h
rts      equ    05h
;
;
;=====
;WAKEUP 暫存器位定義
;=====
wdts     equ    07h
wue      equ    06h
eis      equ    05h
puh3     equ    03h
puh2     equ    02h
puh1     equ    01h
puh0     equ    00h
;
;
;=====
;位置定義
;=====
a_temp   equ    0bh
status_temp equ 0ch
portb_buf equ 0dh
;
;
;=====
;portb 埠      I/O 定義
;=====
wave     equ    00h
;
;
;=====
;=====
;program start here
;=====
;=====
org      3ffh      ;重定向量地址
goto    main
org      3feh      ;中斷副程式入埠位址
goto    time_int
org      000h      ;程式在 ROM 中的起始地址
;
;
;=====
;MAIN PROGRAM
;=====
main:;初始化
        clrm     portb      ;portb 埠輸出"00h"
        clrm     portb_buf ;portb 埠輸出緩衝器清零

```

```

cra
iodir   portb           ;portb 埠設為輸出
movla   b'00000010'
select                ;tmr0 採用內部時鐘計數,預除頻器分配給 tmr0,
                    ;除頻比為 1:8

movla   b'10000001'
movam   irqm           ;開放 tmr0 中斷功能
movla   06h
movam   tmr0          ;給 tmr0 預設值"06h"(6),使 tmr0 計數 250 即溢出
;
;
start::主迴圈
    nop
    nop
    nop
    nop
    nop
    goto  start
;
;
;=====
;tmr0 中斷副程式(每 2mS 中斷次)
;=====
time_int:
    nop                ;時間補償指令
    nop                ;時間補償指令
    movam  a_temp      ;將累加器內容送入 a_temp 中
    swapm  status,w    ;將狀態暫存器 status 高低半位元組內容
                    ;相互交換結果送累加器中
    movam  status_temp ;最後將結果送入 status_temp 中
    movla  08h         ;給 tmr0 初始值"08h"(6+1+1),使
;tmr0 每 2mS((256-6)*8=2000uS)中斷一次,
;為補償 lgoto time_int~movam tmr0 9 條
;指令(耗時 9 個指令週期)及硬體回應中斷所消耗的
;時間(耗時 7 個指令週期),因 tmr0 除頻比為 1:8,給
;tmr0 重裝初值時多加了 2(2=16/8)個計數值
    movam  tmr0
    comm   portb_buf,f
    movm   portb_buf,w
    andla  01h
    movam  portb
end_int:
    bcm   irqf,tmr0f   ;清 tmr0 中斷請求標誌
    swapm  status_temp,w ;將 status_temp 暫存器高低半位元組內容相互交換
                    ;結果送累加器中
    movam  status      ;累加器內容送狀態暫存器 status,以恢復進中

```

```

;斷前的 status 狀態
swapm a_temp,f ;將 a_temp 的高低半位元組內容相互交換,
;結果送入 a_temp 中
swapm a_temp,w ;將 a_temp 的高低半位元組內容相互交換,
;結果送入累加器中,以恢復進中斷前的累加器內容
reti ;中斷返回，執行該指令後，程式將自動開中斷
;
;
;=====
end ;程式結束
;
;
;=====

```

;範例

```

;=====
;系統上電後會在 portb,0 埠輸出一個週期為 4mS 的方波,該
;方波是通過 tmr0 中斷來實現的,tmr0 每 2mS 中斷一次.由於
;中斷相應過程以及在中斷副程式中對累加器,狀態暫存器的保
;護和 tmr0 的重裝初值都需要消耗時間(該部分時間共為 16 個指
;令週期),故副程式中對此部時間作了相應補償,補償方式為:
;因除頻比為 1:8,故給 tmr0 多裝 2 個初值,使其少記 2 個數。
;=====

```

```

;=====
;
;
;模式 (mode):高級 (advanced)
;晶振：4MHZ
;看門狗 (WATCHDOG)：禁止 (DISABLE)
;功能:tmr0 4mS 中斷程式
;系統利用 tmr0 中斷產生一個週期為 8mS 的方波,該放波信號
;通過 portb,0 埠輸出
;
;
;=====

```

;特殊功能暫存器定義

```

;=====
indf equ 00h
tmr0 equ 01h
pc equ 02h
status equ 03h
fsr equ 04h
porta equ 05h
portb equ 06h
wakeup equ 20h
irqm equ 21h
irqf equ 22h

```

```

;
;
;
;=====
;目的暫存器控制位元定義
;=====
w          equ    00h
f          equ    01h
;
;
;=====
;狀態暫存器(status)位元定義
;=====
c          equ    00h
dc         equ    01h
z          equ    02h
pd         equ    03h
to         equ    04h
pa0        equ    05h
pa1        equ    06h
pa2        equ    07h
;
;
;=====
; 中斷遮罩暫存器(irqm)位元定義
;=====
intm       equ    07h
exintm     equ    02h
tmr0m      equ    00h
;
;
;=====
;中斷請求標誌暫存器 (irqf)位元定義
;=====
tmr0f      equ    00h
exintf     equ    02h
;
;
;=====
;select 暫存器位元定義
;=====
ps0        equ    00h
ps1        equ    01h
ps2        equ    02h
psa        equ    03h
rte        equ    04h
rts        equ    05h
;
;
;=====
;WAKEUP 暫存器位元定義

```

```

;=====
wdts      equ    07h
wue       equ    06h
eis       equ    05h
puh3     equ    03h
puh2     equ    02h
puh1     equ    01h
puh0     equ    00h
;
;
;=====
;位置定義
;=====
a_temp    equ    0bh
status_temp equ 0ch
portb_buf equ 0dh
;
;
;=====
;portb 埠    I/O 定義
;=====
wave      equ    00h
;
;
;=====
;program start here
;=====
org       3ffh      ;重定向量地址
goto     main
org       3feh      ;中斷副程式入埠位址
goto     time_int
org       000h      ;程式在 ROM 中的起始地址
;
;
;=====
;MAIN PROGRAM
;=====
main:;初始化
    clrm    portb      ;portb 埠輸出"00h"
    clrm    portb_buf  ;portb 埠輸出緩衝器清零
    clra
    iodir   portb      ;portb 埠設為輸出
    movla  b'00000011'
    select          ;tmr0 採用內部時鐘計數,預除頻器分配給 tmr0,
                   ;除頻比為 1:16
    movla  b'10000001'
    movam  irqm        ;開放 tmr0 中斷功能

```

```

movla 06h
movam tmr0 ;給 tmr0 預設值"06h"(6),使 tmr0 計數 250 即溢出
;
;
start::主迴圈
nop
nop
nop
nop
nop
goto start
;
;
;=====
;tmr0 中斷副程式(每 4mS 中斷次)
;=====
time_int:
nop ;{時間補償指令
nop ;時間補償指令}
movam a_temp ;將累加器內容送入 a_temp 中
swapm status,w ;將狀態暫存器 status 高低半位元組內容
;相互交換結果送累加器中
movam status_temp ;最後將結果送入 status_temp 中
movla 07h ;給 tmr0 初始值"07h"(6+1),使
;tmr0 每 1mS((256-6)*16=4000uS)中斷一次,
;為補償前面幾條指令及相應中斷所消耗的
;時間,給 tmr0 初值時多加了 1 個計數值
movam tmr0
comm portb_buf,f
movm portb_buf,w
andla 01h
movam portb
end_int:
bcm irqf,tmr0f ;清 tmr0 中斷請求標誌
swapm status_temp,w ;將 status_temp 暫存器高低半位元組內容相互交換
;結果送累加器中
movam status ;累加器內容送狀態暫存器 status,以恢復進中
;斷前的 status 狀態
swapm a_temp,f ;將 a_temp 的高低半位元組內容相互交換,
;結果送入 a_temp 中
swapm a_temp,w ;將 a_temp 的高低半位元組內容相互交換,
;結果送入累加器中,以恢復進中斷前的累加器內容
reti ;中斷返回,執行該指令後,程式將自動開中斷
;
;
;=====

```

```

end                ;程式結束
;
;
;=====
;範例
;=====
;系統上電後會在 portb,0 埠輸出一個週期為 8mS 的方波,該
;方波是通過 tmr0 中斷來實現的,tmr0 每 4mS 中斷一次.由於
;中斷相應過程以及在中斷副程式中對累加器,狀態暫存器的保
;護和 tmr0 的初值都需要消耗時間(該部分時間共為 16 個指
;令週期),故副程式中對此部時間作了相應補償,補償方式為:
;因除頻比為 1:16,故給 tmr0 多裝 1 個初值,使其少記 1 個數。
;=====
;
;
;
;模式 (mode):高級 (advanced)
;晶振 : 32.768KHZ
;看門狗 (WATCHDOG) : 禁止 (DISABLE)
;功能:tmr0 1S 中斷程式
;系統利用 tmr0 中斷產生一個週期為 2S 的方波,該方波信號
;通過 portb,0 埠輸出
;
;
;=====
;特殊功能暫存器定義
;=====
indf            equ    00h
tmr0            equ    01h
pc              equ    02h
status          equ    03h
fsr             equ    04h
porta           equ    05h
portb           equ    06h
wakeup         equ    20h
irqm            equ    21h
irqf            equ    22h
;
;
;=====
;目的暫存器控制位元定義
;=====
w              equ    00h
f              equ    01h
;
;
;=====
;狀態暫存器(status)位元定義
;=====

```

```

c          equ    00h
dc         equ    01h
z          equ    02h
pd         equ    03h
to         equ    04h
pa0        equ    05h
pa1        equ    06h
pa2        equ    07h
;
;
;

```

```

;=====
; 中斷遮罩暫存器(irqm)位定義
;=====

```

```

intm       equ    07h
exintm     equ    02h
tmr0m      equ    00h
;
;
;

```

```

;=====
;中斷請求標誌暫存器 (irqf)位元定義
;=====

```

```

tmr0f      equ    00h
exintf     equ    02h
;
;
;

```

```

;=====
;select 暫存器位定義
;=====

```

```

ps0        equ    00h
ps1        equ    01h
ps2        equ    02h
psa        equ    03h
rte        equ    04h
rts        equ    05h
;
;
;

```

```

;=====
;WAKEUP 暫存器位定義
;=====

```

```

wdts       equ    07h
wue        equ    06h
eis        equ    05h
puh3       equ    03h
puh2       equ    02h
puh1       equ    01h
puh0       equ    00h
;
;
;

```

```

;=====
;位置定義

```

```

;=====
a_temp      equ    0bh
status_temp equ    0ch
portb_buf   equ    0dh
;
;
;=====
;portb 埠    I/O 定義
;=====
wave        equ    00h
;
;
;=====
;=====
;program start here
;=====
;=====
org    3ffh      ;重定向量地址
goto   main
org    3feh      ;中斷副程式入埠位址
goto   time_int
org    000h      ;程式在 ROM 中的起始地址
;
;
;=====
;MAIN PROGRAM
;=====
main:      ;初始化
    clrm    portb      ;portb 埠輸出"00h"
    clrm    portb_buf   ;portb 埠輸出緩衝器清零
    clra
    iodir   portb      ;portb 埠設為輸出
    movla   b'00000100'
    select  ;tmr0 採用內部時鐘計數,預除頻器分配給 tmr0,
            ;除頻比為 1:32

    movla   b'10000001'
    movam   irqm        ;開放 tmr0 中斷功能
    movla   00h
    movam   tmr0        ;給 tmr0 預設值"00h"(0),使 tmr0 計數 256 即溢出
;
;
;
start::;主迴圈
    nop
    nop
    nop
    nop
    goto   start


```

```


;
;
;=====
;tmr0 中斷副程式(每 1mS 中斷次)
;=====
time_int:
    movam  a_temp      ;將累加器內容送入 a_temp 中
    swapm  status,w    ;將狀態暫存器 status 高低半位元組內容
                        ;相互交換結果送累加器中
    movam  status_temp ;最後將結果送入 status_temp 中
    comm   portb_buf,f
    movm   portb_buf,w
    andla  01h
    movam  portb
end_int:
    bcm    irqf,tmr0f  ;清 tmr0 中斷請求標誌
    swapm  status_temp,w ;將 status_temp 暫存器高低半位元組內容相互交換
                        ;結果送累加器中
    movam  status      ;累加器內容送狀態暫存器 status,以恢復進中
                        ;斷前的 status 狀態
    swapm  a_temp,f    ;將 a_temp 的高低半位元組內容相互交換,
                        ;結果送入 a_temp 中
    swapm  a_temp,w    ;將 a_temp 的高低半位元組內容相互交換,
                        ;結果送入累加器中,以恢復進中斷前的累加器內容
    reti                               ;中斷返回，執行該指令後，程式將自動開中斷
;
;
;=====
    end                               ;程式結束
;
;=====
;範例
;=====
;系統上電後會在 portb,0 埠輸出一個週期為 2S 的方波,該
;方波是通過 tmr0 中斷來實現的,tmr0 每 1S 中斷一次.tmr0
;除頻比為 1:32,時鐘採用 32.768KHZ 故 tmr0 溢出一即
;為 1S(1000000/32768*4*256*32uS/1000000=1S)

```

應用線路圖



Title: TM58XX TMRD 100ms 1ms 2ms 4ms 中斷 APN 線路圖

	Size: B	Number:	Revision:
	Date:		Shastof
	File:		Drawn by:

